

Devantech SRF04 Ultrasonic Range Finder (#28015)

The Devantech SRF04 ultrasonic range finder provides precise, non-contact distance measurements from about 3 cm (1.2 inches) to 3 meters (3.3 yards). It is very easy to connect to BASIC Stamps or the Javelin, requiring only two I/O pins.¹ The SRF04 library makes this device very simple to use and is an ideal component for robotics applications.

The SRF04 works by transmitting an ultrasonic (well above human hearing range) pulse and measuring the time it takes to "hear" the pulse echo. Output from the SRF04 is in the form of a variable-width pulse that corresponds to the distance to the target.

The SRF04 is designed and manufactured by Devantech, who provides additional technical resources for the device. Their web site is <http://www.robot-electronics.co.uk>.


Features

- Voltage – 5 v
- Current – 30 mA Typ. 50mA Max.
- Frequency – 40 kHz
- Max Range - 3 m
- Min Range - 3 cm
- Sensitivity - Detect 3 cm diameter broom handle at > 2 m
- Input Trigger – 10 uS Min. TTL level pulse
- Echo Pulse - Positive TTL level signal, width proportional to range.
- Small Size – (1.7 in x .8 in x .7 in height) 43 mm x 20 mm x 17 mm height


¹ For a Javelin Stamp application note see www.javelinstamp.com.

Connection to the BASIC Stamp 2

The SRF04 has four through-hole locations where you will need to solder wires to the hardware. These wires are not included with the kit.




SRF04 Connections



Source Code Example

The SRF04 detects objects by emitting a short burst of sound and "listening" for the echo. Under control of the BASIC Stamp, the SRF04 emits an ultrasonic (40 kHz) sound pulse. This pulse travels through the air at about 1.125 feet per millisecond (the speed of sound), hits an object and then bounces back. By measuring the time between the transmission of the pulse and the echo return, the distance to the object can be determined.



The SRF04 outputs a high-going pulse that corresponds to time required for the echo to return. **RCTIME** (for the BS2 and BS2e) and **PULSIN** (for the BS2sx, BS2p, and BS2pe) can be used to measure it and determine the distance to the target. Using **RCTIME** on the BS2 and BS2e allows us to measure the echo

pulse without "seeing" the low-to-high transition of the echo's leading edge – something we may miss due to the setup time for `PULSIN` on the BS2 and BS2e.


There are a couple of small technical details to be aware of, but otherwise, coding for the SRF04 is very straight forward. The trigger pulse must be at least 10 microseconds long. `PULSOUT` can do this for us. The other requirement is that we must wait at least 10 milliseconds between measurements.

The heart of this program is a subroutine called `Get_Sonar`. This routine started with the code sample that came with the sensor. While the Devantech example is perfectly suitable and easy to understand, the results seemed to bounce around a bit. Some software filtering (averaging several readings together) would smooth things out and make the output more useful.

The use of `RCTIME` versus `PULSIN` for the echo timing measurement is determined at compile time by the BASIC Stamp Editor/Compiler. The structure of the program is such that it may be used with any BASIC Stamp module. If the `$STAMP` directive does not match the module you're attempting to program, the editor will assist in correcting this declaration for you.

The value returned by the routine will be stored in `rawDist`, so the code starts by clearing it. Then, within a loop, the code takes five readings from the sensor and averages them together. This may look a bit odd because we usually think about adding numbers and then dividing to get an average. We do the dividing first and then add the result into the return value because we could have an overflow if we do all the addition first. Yes, the dividing first technique can lead to rounding errors, but only if the values were very small. We didn't observe this when using the SRF04. Since one inch (the minimum range of the sensor) is about 74 microseconds, dividing by five (loop value) each time through causes no problem.

Once returned to the main program the `rawDist` value is converted to centimeters by dividing the echo duration by the appropriate factor (`ToCm`) – this factor determined by the BASIC Stamp module installed. Conversion from centimeters to inches is straightforward math; in this case the centimeters value is multiplied by 3.937 to convert to tenths of inches.



```

' =====
'
' File..... SRF04_Demo.BS2
' Purpose.... Devantech SRF04 Ultrasonic Range Finder
' Author..... Parallax, Inc. (Copyright 2003 - All Rights Reserved)
' E-mail..... support@parallax.com
' Started.... 06 MAR 2002
' Updated.... 01 OCT 2003
'
'   {$STAMP BS2}
'   {$PBASIC 2.5}
'
' =====

' -----[ Program Description ]-----
'
' This program uses the Devantech SRF04 to measure the distance between the
' unit and a target.  Display is raw value, centimeters, and inches.

' -----[ Revision History ]-----
'
' 01 OCT 2003 : Updated for PBASIC 2.5 and for any BASIC Stamp module

' -----[ I/O Definitions ]-----
'
Trigger          PIN      0
Echo             PIN      1

' -----[ Constants ]-----

#SELECT $STAMP
#CASE BS2, BS2E
    Trig10      CON      5           ' trigger pulse = 10 uS
    ToCm        CON      30         ' conversion factor to cm
#CASE BS2SX, BS2P
    Trig10      CON      13
    ToCm        CON      78
#CASE BS2PE
    Trig10      CON      5
    ToCm        CON      31
#ENDSELECT

' -----[ Variables ]-----

samples          VAR      Nib           ' loop counter
pWidth           VAR      Word         ' pulse width from sensor
rawDist          VAR      Word         ' filtered measurment
cm               VAR      Word         ' centimeters
inches           VAR      Word

```

```

' -----[ Initialization ]-----
Setup:
  LOW Trigger
  DEBUG CLS,
    "Devantech SRF04 Demo", CR,
    "-----", CR,
    "Raw.....", CR,
    "Centimeters...", CR,
    "Inches....."

' -----[ Program Code ]-----

Main:
  DO
    GOSUB Get_Sonar           ' take sonar reading
    DEBUG CRSRXY, 15, 2, DEC rawDist, CLREOL

    cm = rawDist / ToCm      ' convert to centimeters
    DEBUG CRSRXY, 15, 3, DEC cm, CLREOL

    inches = cm * / $03EF    ' x 3.937 (to 0.1 inches)
    DEBUG CRSRXY, 15, 4,
      DEC inches / 10, ".", DEC1 inches,
      CLREOL

    PAUSE 250                ' delay between readings
  LOOP

  END

' -----[ Subroutines ]-----

Get_Sonar:
  rawDist = 0                ' clear measurement
  FOR samples = 1 TO 5      ' take five samples
    PULSOUT Trigger, Trig10 ' 10 uS trigger pulse
    #SELECT $STamp
      #CASE BS2, BS2E
        RCTIME Echo, 1, pWidth ' measure pulse
      #CASE #ELSE
        PULSIN Echo, 1, pWidth ' measure pulse
    #ENDSELECT
    rawDist = rawDist + (pWidth / 5) ' simple digital filter
    PAUSE 10                    ' minimum period between
  NEXT
  RETURN

```